

```
// This function allows you to connect an ARRAY with any datatype (Please look
// in the online help of SFC20 for exceptions) in a CFC. You only have to insert
// the source and target area of the block you have inserted in the CFC.
```

```
FUNCTION FC 88 : VOID
```

```
NAME : ARRAY_L
AUTHOR : UNKNOWN
FAMILY : USER
VERSION : 0.1
```

```
VAR_INPUT
    DatenTyp1 : BYTE ;
    Retryfactor1 : WORD ;
    DB_No1 : WORD ;
    Byte_No1 : DINT ;
    DataTyp2 : BYTE ;
    Retryfactor2 : WORD ;
    DB_No2 : WORD ;
    Byte_No2 : DINT ;
END_VAR
```

```
VAR_OUTPUT
    Return_Value : INT ;
END_VAR
```

```
VAR_TEMP
    AnyPointer1 : ANY ;
    AnyPointer2 : ANY ;
END_VAR
```

```
BEGIN
```

```
NETWORK
TITLE =FunctionCall
```

```
LAR1  P##AnyPointer1;           // Load the start address of ANY-Pointers in AR1
L      B#16#10;                 // Syntax-ID (Step7 has everytime 10H)
T      LB [AR1,P#0.0];
L      #DataTyp1;               // Load datatype
T      LB [AR1,P#1.0];
L      #Retryfactor1;           // Length of area (Datatyp, Word and
                                // Retryfactor=10 -> 10 * 2 Bytes (Length
                                // of Word) = 20Byte)

T      LW [AR1,P#2.0];
L      #DB_No1;                 // Number of DB
T      LW [AR1,P#4.0];
L      #Byte_No1;               // Number of Byte in the DB
SLD    3;
T      LD [AR1,P#6.0];
L      B#16#84;                 // Adressing a Data Block
T      LB [AR1,P#6.0];
```

```

LAR1  P##AnyPointer2;          // Load the start address of ANY-Pointers in AR1
L      B#16#10;                // Syntax-ID (Step7 has everytime 10H)
T      LB [AR1,P#0.0];
L      #DataTyp2;              // Load datatype
T      LB [AR1,P#1.0];
L      #Retryfactor2;          // Length of area (Datatyp, Word and
                                // Retryfactor=10 -> 10 * 2 Bytes (Length
                                // of Word) = 20Byte)

T      LW [AR1,P#2.0];
L      #DB_No2;                // Number of DB
T      LW [AR1,P#4.0];
L      #Byte_No2;              // Number of Byte in the DB
SLD    3;
T      LD [AR1,P#6.0];
L      B#16#84;                // Addressing a Data Block
T      LB [AR1,P#6.0];

CALL "BLKMOV" (SRCBLK:= #AnyPointer1,RET_VAL:= #Return_Value,DSTBLK:= #AnyPointer2);

END_FUNCTION

```